

A BASIC Computer Program to Calculate Daily Potential Evapotranspiration

Douglas M. Stone

Research forester, USDA Forest Service, North Central Forest Experiment Station, Grand Rapids, MN

A BASIC microcomputer program was developed to calculate daily potential evapotranspiration (PET) by the method of Thornthwaite and Mather. Daily and cumulative PET values are calculated and displayed as rapidly as daily air temperatures are entered. In example, PET is used as an index of actual evapotranspiration where soil water is maintained at low tensions. Tree Planters' Notes 39(3): 9-12; 1988.

In 1948, Thornthwaite (3) introduced the concept of potential evapotranspiration (PET)—the water lost from a fully vegetated site that is well supplied with soil water. Although PET is determined by many climatic factors, it can be estimated from average air temperature and day length using a method developed by Thornthwaite and Mather (4). This method has been applied over a wide range of forest environments. Zahner (7) first used it to calculate monthly and growing season soil water deficits of southern pine sites across the mid-South.

In Alaska, where tree growth is frequently limited by temperature, Patric and Black (2) reported that PET reflected forest distribution better than temperature alone. Verry and Timmons (6) found that the PET of a complex upland-peatland watershed

in northern Minnesota approximated the water balance value within 8% on an annual basis and within 3% over a 3-year period of average to wet years. First-year survival of pine seedlings planted from 1929 to 1976 on droughty sands in Michigan was significantly related to growing season soil water deficits calculated as the difference between monthly precipitation and PET (1).

A BASIC microcomputer program to estimate PET by the Thornthwaite and Mather method was developed as part of a study to determine nutrient leaching in a sandy forest soil treated with simulated acid rain. Two-year-old white spruce (*Picea glauca* (Moench) Voss) seedlings are being grown in 15 x 30-cm soil columns equipped with porous polyethylene filters to collect leachate passing through the root zone. The seedlings are covered at night and during periods of rain to exclude dew and natural rainfall. The soil cores are treated weekly; the leachate is collected, weighed, and analyzed for 20 chemical characteristics. About 100 ml of leachate is required for the analyses. We use weekly PET values to estimate how much water to apply to the soil columns to obtain a sufficient sample and to avoid excessive leaching due to overwatering.

Data Files

To calculate PET, two data files are required—one for unadjusted PET and one for daylength. Values for each are given in Thornthwaite and Mather (5). The average monthly temperature for the location must be known to determine the proper values. Figure 1 lists a BASIC program to enter the unadjusted PET values into a data file called "UPET." Comments following apostrophes provide information for the user but are ignored by the computer. The statement at line 150 opens the file and the loop at lines 160-230 enters the data for temperatures from 34 to 80 °F in 0.5° increments.

PET at temperatures below 34 °F is negligible for most areas; values for average daily temperatures above 80°F are available in Thornthwaite and Mather (5). It is quicker to simply enter the numbers and let the computer place the decimal (line 180); the values are displayed at line 190. The optional check at line 200 requires an answer for each entry; however, this may be quicker than manually checking and correcting errors after the file is created.

Figure 2 lists a program to generate the day-length file; it works in the same way as the previous one. Day-length values in 12-hour units (without the decimal) are entered for each

```

100 'TEMPET.BAS      D. M. STONE      12-2-86
110 '
120 'TO LOAD TEMP & PET DATA IN "UPET" FILE
130 '(THORNTHWAITE & MATHER 1957, TABLE 3.)
140 '
150 OPEN "O", #1, "UPET"
160 FOR T = 34 TO 80 STEP .5
170   INPUT "ENTER THE PET VALUE";PE
180   P = PE/100
190   PRINT, T,P
200   INPUT "IS THE PET CORRECT, (Y OR N)"; AS
210   IF AS = "N" THEN 170
220   PRINT #1, T;P
230   NEXT T
240 CLOSE #1
250 END

```

Figure 1—A BASIC program to create the unadjusted PET (UPET) file.

```

100 'SUNLITE.BAS    D. M. STONE      12-2-86
110 '
120 'TO LOAD DAY LENGTH DATA IN "DAYLT" FILE
130 '(THORNTHWAITE & MATHER 1957, TABLE 8.)
140 '
150 OPEN "O", #1, "DAYLT"
160 FOR J = 1 TO 365      '(J = JULIAN DATE)
170   INPUT "ENTER THE DAY LENGTH"; DL
180   D = DL/100
190   PRINT, J,D
200   INPUT "IS THE DAY LENGTH CORRECT, (Y OR N)"; AS
210   IF AS = "N" THEN 170
220   PRINT #1, J;D
230   NEXT J
240 CLOSE #1
250 END

```

Figure 2—A BASIC program to generate the day-length (DAYLT) file.

day of the year. The latitude of the site must be known to enter the appropriate values for duration of sunlight. For sites in northern latitudes, this program could be modified to exclude the winter months, provided the appropriate Julian dates are used. If, for example, you wish to use an April through October season, statement 160 should read "FOR J = 91 to 304." This shortened season would require less memory, and its file would load more quickly. We used the data for the entire year so that we could evaluate the PET program under greenhouse conditions during the winter.

It takes about an hour to create the unadjusted PET and the day-length files; they then can be copied onto a backup disk and used indefinitely. The two files combined occupy less than 6,000 bytes of memory.

The Program

The program calculates and displays daily PET values for 7 days in just over a minute or values for 1.4 days in less than 2 minutes. The calculations are performed as fast as the data are entered. It takes about 6 seconds to load the two data files into the memory. The program and files combined require less than 8,000 bytes.

Figure 3 is a listing of the program. The dimension statements (DIM) on lines 330 and 340

```

100 "PET"      D. M. STONE      12-2-86
110
120 TO CALCULATE DAILY POTENTIAL EVAPOTRANSPIRATION (PET)
130 BY THE THORNTHWAITE & MATHER (1957) METHOD.
140
150
160 .....
170      VARIABLE NAMES:
180 .....
190      J = JULIAN DATE
200      N = NUMBER OF DAYS
210      P = UNADJUSTED PET
220      T = AVERAGE DAILY TEMPERATURE (F)
230      DL = DAYLENGTH
240      TN = MINIMUM DAILY TEMPERATURE
250      TX = MAXIMUM DAILY TEMPERATURE
260      PET = DAILY POTENTIAL EVAPOTRANSPIRATION
270      TPET = TOTAL PET FOR THE PERIOD
280      UPET = UNADJUSTED PET FILE NAME
290      DAYLT = DAY LENGTH FILE NAME
300 .....
310 .....
320
330 DIM A(99)
340 DIM B(370)
350 OPEN "I", #1, "UPET"
360 WHILE NOT EOF (1)
370     INPUT #1, T,P
380     A(T) = P
390 WEND
400
410 OPEN "I", #2, "DAYLT"
420 WHILE NOT EOF (2)
430     INPUT #2, J,DL
440     B(J) = DL
450 WEND
460
470 LET TPET = 0
480 INPUT "ENTER THE JULIAN DATE OF THE FIRST DAY AND NO. OF DAYS.":J,N
490 PRINT
500 FOR I = 1 TO N
510     INPUT "ENTER THE MAX. & MIN. TEMP.": TX,TN
520     PRINT
530     PRINT "DAY NO.",I,"MAX.",TX,"MIN.", TN
540     INPUT "ARE ALL ENTRIES CORRECT, (Y OR N)":AS
550     IF AS = "N" THEN 510
560     T = (TX + TN)/2
570     PET = A(T) * B(J)
580     TPET = TPET + PET
590     PRINT,"PET FOR DAY":I,"=":PET,"TOTAL FOR THE PERIOD IS:":TPET
600     PRINT
610     B(J) = B(J + 1)
620     NEXT I
630
640 PRINT
650 PRINT "THE ESTIMATED PET FOR THIS":N,"-DAY PERIOD IS:":TPET," In."
660 PRINT
670 INPUT "DO YOU WANT TO MAKE ANOTHER RUN, (Y OR N)": AS
680 IF AS = "Y" THEN 470
690 CLOSE #1 : CLOSE #2
700 END

```

reserve space in arrays A and B for the two data files; lines 350 and 410 open them for input to the program. The loop on lines 360-390 loads the temperature and unadjusted PET values from the "UPET" file into memory; that on lines 420-450 loads the Julian dates and corresponding day-length values. The PET accumulator is set to zero in 470. At 480 the beginning date and number of days in the run are entered; they must be separated by a comma. The main loop begins at 500; in 510 the maximum and minimum temperatures for the first day are entered (separated by a comma). Print statements display a blank line on the monitor and make it easier to read. Line 530 displays the day number and corresponding temperatures; 540 is an optional check to ensure that the correct values were entered. If they were, the "Y" key is pressed to continue; if not, the "N" is pressed and the program returns to 510 so that the correct values can be entered. At 560 the average temperature for the day is calculated, and in 570 the PET (the product of the values in the two arrays). The daily PET is added to the total in 580, and both are displayed at 590. The Julian date is incremented in 610, and 620 transfers control back to 510 to enter the temperatures for the next day. After the loop is completed for

Figure 3—Listing of a BASIC program to calculate daily PET.

the number of days specified in 480, the results are displayed in 650. If additional calculations are desired, pressing the "Y" key at 670 transfers control back to 470; otherwise, the data files are closed and the program ends.

Discussion

PET data have been used mainly in conjunction with precipitation measurements to estimate water balances and/or soil water deficits for monthly (7), growing season (1), annual (6), or longer periods (2). PET also can be used to estimate actual evapotranspiration (AET) for short intervals because the two agree closely as long as soil water is readily available; that is, at tensions up to about 2 atmospheres (8). For the spruce seedlings described earlier, we can usually estimate weekly AET within 50 ml, or about 0.1 inch. PET is thus a useful index of AET, enabling us to predict the water needed to recharge the soil columns and obtain the required volume of leachate for analyses.

Equally important, it helps avoid overwatering.

Thornthwaite's method of calculating PET has been widely applied because it provides good estimates, requires only average daily air temperatures, and is easily calculated. Daily or periodic PET estimates could be useful in forest nurseries or in other environments where soil water is maintained at low tensions. Estimates of PET may be especially useful where overwatering could cause excessive leaching or other problems and where pumping costs must be minimized.

Copies of the program may be obtained by sending a formatted 360-K, 5¼-inch diskette to Douglas M. Stone, USDA Forest Service, Forestry Sciences Laboratory, 1831 Highway 169 East, Grand Rapids, MN 55744.

Literature Cited

1. Cleland, D.T.; Johnson, J.E. Pine plantation survival related to calculated moisture deficits on the Huron National Forest (1929-1976). *Tree Planters' Notes* 37:17-22; 1986.
2. Patric, J.H.; Black, P.E. Potential evapotranspiration and climate in Alaska by Thornthwaite's classification. Res. Paper PNW-71. Portland, OR: U.S. Department of Agriculture, Forest Service, Pacific Northwest Forest and Range Experiment Station; 1968. 28 p.
3. Thornthwaite, C.W. An approach toward a rational classification of climate. *Geographical Review* 38:55-94; 1948.
4. Thornthwaite, C.W.; Mather, J.R. The water balance. Drexel Institute of Technology, Laboratory of Climatology, Publications in Climatology 8(1): 1-86; 1955.
5. Thornthwaite, C.W.; Mather, J.R. Instructions and tables for computing potential evaporation and the water balance. Drexel Institute of Technology, Laboratory of Climatology, Publications in Climatology 10(3): 1-311; 1957.
6. Verry, E.S.; Timmons, D.R. Waterborne nutrient flow through an upland-peatland watershed in Minnesota. *Ecology* 63:1456-1467; 1982.
7. Zahner, R. Evaluating summer water deficiencies. Occ. Paper 150. New Orleans, LA: U.S. Department of Agriculture, Forest Service, Southern Forest Experiment Station; 1956. 18 p.
8. Zahner, R. Refinement in empirical functions for realistic soil-moisture regimes under forest cover. In: Sopper, W.E.; Lull, H.W., ed. *Forest hydrology*. New York: Pergamon; 1967: 261-274.